Q:

Excel macro to get SEPARATE graphs of many Ys with the same Xs

I have a data set in which the first column is the x but the other columns (> 50) are the Ys. There are 6 rows. I know I can plot all the Ys on the same axis in the same chart, but I need separate graphs of Y all with the same x. However, I am clueless if it is possible to have an excel macro that will allow me to plot DIFFERENT bar graphs  for as many Ys all at once. Doing it manually is a pain because I have many datasets with these dimensions. 
Any help would be greatly appreciated. Below is the code I used but it plotted all the bars in a single graph. Found this code online and tweaked it
Sheets("sheet1").Select
 ActiveSheet.Shapes.AddChart.Select
 With ActiveChart
 .ChartType = xlColumnClustered
.SetSourceData Source:=Sheets("sheet1").Range("A1:D7")  'sets source  data for graph including labels
.SetElement (msoElementLegendRight)  'including legend
.HasTitle = True
'dimentions & location:
.Parent.Left = 47   'defines the coordinates for the left side of the chart
.Parent.Height = 300
.Parent.Width = 600
.ChartTitle.Text = "Yield by group " & intGraphStart
End With

A sample

A:

The procedure below create a chart in a separated sheet for each of the columns from B to L using column A as a common horizontal axis.
Sub Rng_Charts_Add_Multiple()
Const kChrTtl As String = "Yield by group "
Dim Wbk As Workbook, WshSrc As Worksheet
Dim rSrc As Range, Chrt As Chart
Dim sCol As String, bCol As Byte

    Rem Application Settings
    Application.EnableEvents = False
    Application.DisplayAlerts = False
    Application.ScreenUpdating = False

    Set Wbk = ThisWorkbook
    Set WshSrc = Wbk.Sheets("Sht(1)")
    Application.Goto WshSrc.Cells(1), 1
    Set rSrc = WshSrc.Cells(1).CurrentRegion

    For bCol = 2 To rSrc.Columns.Count

        Rem Add Chart in a New Sheet
        sCol = rSrc.Cells(1, bCol).Value2
        With Wbk
            On Error Resume Next
            .Sheets("Chart " & sCol).Delete
            On Error GoTo 0
            Application.Goto WshSrc.UsedRange.SpecialCells(xlCellTypeLastCell).Offset(2, 2)
            Set Chrt = .Charts.Add2(After:=.Sheets(.Sheets.Count), NewLayout:=1)
        End With

        Rem Chart Settings
        With Chrt
            .name = "Chart " & sCol
            .ApplyLayout (1)
            .ChartTitle.Text = kChrTtl & sCol
            .ChartType = xlColumnClustered
            .SetSourceData Source:=Union(rSrc.Columns(1), rSrc.Columns(bCol))

    End With: Next

    Rem Application Settings
    Application.EnableEvents = True
    Application.DisplayAlerts = True
    Application.ScreenUpdating = True

End Sub

Suggest to visit the following pages:
Variables & Constants, Application Object (Excel), Excel Objects
With Statement, For...Next Statement, Range Object (Excel)
Worksheet Object (Excel), ChartObjects Members (Excel)

